

А.П.Ростов

УНИВЕРСАЛЬНЫЙ МНОГОКАНАЛЬНЫЙ АНАЛОГО-ЦИФРОВОЙ ПРЕОБРАЗОВАТЕЛЬ ДЛЯ ЭКСПЕРИМЕНТАЛЬНЫХ ИССЛЕДОВАНИЙ АТМОСФЕРЫ

Предложен вариант 10-разрядного аналого-цифрового преобразователя, имеющего интерфейс RS-232, не требующего специального оборудования для стыковки с любым персональным компьютером и пригодного для приборов и установок, используемых в международных экспериментах.

В последнее время все чаще проводятся совместно с зарубежными коллегами сложные натурные эксперименты в разных странах и на различных континентах. При этом возникают сложности при стыковке привозимой нашими учеными аппаратуры с разными типами компьютеров, используемых в эксперименте.

Поскольку практически все разрабатываемые в настоящее время приборы имеют в своем составе аналого-цифровой преобразователь (АЦП), то для решения этой проблемы автором данной статьи был разработан многоканальный АЦП, имеющий международный стандартный интерфейс RS-232 [1]. Такой интерфейс имеют все современные компьютеры. Для нормальной работы по протоколу этого интерфейса требуется некоторая <сообразительность> АЦП, желательно было бы добавить к нему микропроцессор. Лучше всего для этих целей подходят однокристальные микроконтроллеры *i 80196* или *i 8051GB* фирмы Intel, имеющие в своем составе 8-канальный АЦП и последовательный порт. Это хорошие современные, но очень дорогостоящие микроконтроллеры, а у разработчиков электронной техники имеются в наличии обычно первые отечественные однокристальные микроконтроллеры и АЦП выпуска восьмидесятых годов. С учетом вышесказанного предлагается вариант дешевого прибора, изготовленного на таких микросхемах. На рис. 1 и 2 приведена принципиальная схема.

Динамический диапазон аналоговых сигналов, которые необходимо оцифровать, редко превышает 60 дБ, и в большинстве случаев достаточно десятиразрядного АЦП. Хорошо для этой цели подходит однокристальный АЦП K1113ПВ1А [2], который является аналогом прибора AD571 фирмы Analog Devices Inc. Этот прибор выпускался в свое время Рижским объединением <Альфа>. Время преобразования менее 30 мкс достаточно, поскольку интерфейс RS-232 накладывает ограничение по скорости передачи информации. При скорости обмена 9600 бот/с реальный поток передачи информации составляет порядка 800 байт/с.

Для управления работой АЦП, приемом и передачей информации и некоторой ее обработкой используется однокристальный микроконтроллер KP1816BE35 [3], который является аналогом прибора *i 8035* фирмы Intel. В свое время он выпускался предприятиями электронной промышленности бывшего СССР. Этот микрокомпьютер имеет внешнюю память программ (ВПМ), что дает возможность простой заменой постоянного запоминающего устройства (ПЗУ) изменить режим работы АЦП и алгоритмы предварительной обработки сигналов. Один из вариантов управляющей программы, написанной на языке Ассемблера ASM-48, приведен ниже.




Рис. 1. Принципиальная схема АЦП (начало)




Рис. 2. Принципиальная схема АЦП (окончание)

```

=====
;Программа управления АЦП-10
;с микроконтроллером КР1816ВЕ35
-----
; А.Ростов, Томск, E-mail andrey@rostov.tomsk.su
-----
;Вывод в стандарте RS-232 (8N1)
;Первым идет маркер AA55H, затем - три двухбайтовых слова,
;содержащие значение трех входных параметров
-----
;Порт 1 ввод младшего байта АЦП
;Порт 2(4p,5p) ввод старших разрядов АЦП
;Порт 2(6p) флаг готовности АЦП
;Порт 2(7p) выход RS-232

Volum0 .EQU 32
Volum1 .EQU 34
Volum2 .EQU 36
Volum3 .EQU 38

.ORG 0
JMP Start

.ORG 100h

;подпрограмма задержки 100 mks использует R7
Delay1: mov R7,#10
Loop1: djnz R7,Loop1
ret

;подпрограмма вывода байта из аккумулятора в порт P2.7
Out232:
mov R6,#8
anl P2,#01111111B ;Start bit
call Delay1
Rotate: rrc A
jc OFF
ON: anl P2,#01111111B
jmp Delay
OFF: orl P2,#10000000B
jmp Delay
Delay: call Delay1
djnz R6,Rotate
orl P2,#10000000B ;Stop bit
call Delay1
ret

Start: .ORG 200h

;Настроим порты
mov A,#0FFh ;ввод
outl P1,A
anl P2,#10001111B ;вывод (4,5,6,7 разрядов)

Loop:
mov R0,#Volum0+1

; Оцифровка 0 канала
mov A,#11001111B
outl P2,A
anl P2,#10111111B

Ready 0: ins A,BUS
jb7 Ready 0
ins A,BUS
anl A,#3 ;наложим маску информация 1 и 2 битах
mov @R0,A
in A,P1 ;считаем младший байт АЦП
dec R0
mov @R0,A
inc R0
inc R0
inc R0 ;сместим указатель на следующий отсчет

```

```

; Оцифровка 1 канала
    orl    P2,#01010000B    ;оцифровка 1 канала
    anl    P2,#10111111B

Ready1: ins    A,BUS
        jb7    Ready1
        ins    A,BUS
        anl    A,#3        ;наложим маску информация 1 и 2 битах
        mov    @R0,A
        in    A,P1        ;считаем младший байт АЦП
        dec    R0
        mov    @R0,A
        inc    R0
        inc    R0        ;сместим указатель на следующий отсчет

; Оцифровка 2 канала

        mov    A,#11101111B    ;оцифровка 2 канала
        outl   P2,A
        anl    P2,#10111111B

Ready2: ins    A,BUS
        jb7    Ready2
        ins    A,BUS
        anl    A,#3        ;наложим маску информация 1 и 2 битах
        mov    @R0,A
        in    A,P1        ;считаем младший байт АЦП
        dec    R0
        mov    @R0,A
        inc    R0
        inc    R0
        inc    R0        ;сместим указатель на следующий отсчет

Оцифровка 3 канала
    orl    P2,#01110000B    ;оцифровка 3 канала
    anl    P2,#10111111B

Ready3: ins    A,BUS
        jb7    Ready3

        anl    A,#3        ;наложим маску информация 1 и 2 битах
        mov    @R0,A
        in    A,P1        ;считаем младший байт АЦП
        dec    R0
        mov    @R0,A

;выдадим оцифрованные параметры в последовательном виде
    mov    A,#0AAh        ;сначала выдадим маркер
    call   Out232        ;AAh
    mov    A,#55h
    call   Out232        ;55h

    mov    R0,#Volum0
    mov    R2,#6        ;количество выдаваемых байт
Loop2: mov    A,@R0
        call   Out232
        inc    R0
        djnz  R2,Loop2

        jmp    Loop        ;заиклим процесс оцифровки
.END        ;конец программы

```

Для коммутации аналоговых сигналов используется коммутатор КР590КН6, уверенно коммутирующий сигналы амплитудой ± 10 В. Для работы АЦП, коммутатора, приемника и передатчика интерфейса RS-232 требуется напряжение ± 12 В. Для удобства использования в состав прибора был введен преобразователь напряжения с $+ 5$ В на ± 10 В. Таким образом, для питания прибора в целом нужно только одно напряжение $+ 5$ В, при этом потребляемый ток не превышает 500 мА.

Для приема информации от прибора, оснащенного таким АЦП, на персональном компьютере (ПК) в простейшем случае нужно набрать команду COPY, а из языков высокого уровня лучше пользоваться программой-драйвером, написанной на языке Ассемблера. Пример такой программы приведен ниже.

```

;=====
;Подпрограмма AdcCom1 на языке Макроассемблер MASM.EXE
;Для языков программирования Fortran и QuikBasic фирмы Microsoft
;Принимает три 10-разрядных слова с порта COM1 IBM-PC
;-----
; Вызов из QB call AdcCom1 (r3, r2, r1)
; где r1, r2, r3 – целочисленные переменные
;-----
Com1_Dat EQU 3F8h
Com1_St EQU 3Fdh

.MODEL medium
.CODE
;кодовый сегмент

PUBLIC ADCCOM1
ADCCOM1 PROC
;объявим имя процедуры

        push    bp
        mov     bp,sp
;сохраним регистр-указатель BP
;возьмем адрес вершины стека

start:  mov     dx,Com1_St

mark1:  in      al,dx
        and     al,1
        jz     mark1
;проверим есть ли байт в приемнике

        mov     dx,Com1_Dat
;есть, возьмем его

        in      al,dx
        cmp    al,0aah
;начало маркера?
        jnz   start
;нет, то все сначала

        mov     dx,Com1_St

mark2:  in      al,dx
        and     al,1
        jz     mark2
;возьмем 2 байт

        mov     dx,Com1_Dat
        in      al,dx
;проверим его на принадлежность маркеру
        cmp    al,55h
        jnz   start
;нет, все сначала
;маркер принят, примем далее 6 байт информации

rws:    mov     bx,[bp+6]
;вычислим адрес первой переменной

        mov     dx,Com1_St

m1:     in      al,dx
        and     al,1
;ждем 1 байт

        jz     m1
        mov     dx,Com1_Dat
        in      al,dx
;примем его
        mov     ah,al
;запомним его в старшем байте аккумулятора

        mov     dx,Com1_St

m2:     in      al,dx
        and     al,1
        jz     m2
;примем 2 байт

        mov     dx,Com1_Dat
        in      al,dx
;переставим байты аккумулятора
        xchg   al,ah
;занесем результат по адресу первой переменной
        mov     [bx],ax
;увеличим указатель
        inc    bp
;на 2
        inc    bp
        loop   rws
;заиклим процесс приема информации
;до получения 6 байт
;восстановим указатель

        pop     bp
        ret    6
;вернемся в вызывающую программу

ADCCOM1 ENDP
END

```

Конструктивно АЦП выполнен в виде платы размером 100×100 мм. Прибор успешно эксплуатируется в ИОА СО РАН в составе автоматического морского солнечного фотометра. Этот прибор успешно отработал в Российско-американо-испанской экспедиции на борту испанского судна <Esperanza del Mar> весной 1994 г. и летом этого же года в комплексной программе <SATOR-94>. В Институте физики прочности и материаловедения СО РАН такой АЦП используется для оцифровки и ввода в РС сигнала от электронного микроскопа.

1. Мячев А.А., Степанов В.Н., Щербо В.К. Интерфейсы систем обработки данных. М.: Радио и связь, 1989. 416с.
2. Федороков Б.Г., Телец В.А. Микросхемы ЦАП и АЦП: функционирование, параметры, применение. М.: Энергоатомиздат, 1990. 320 с.
3. Сташин В.В., Урусов А.В., Мологонцева О.Ф. Проектирование цифровых устройств на однокристальных микроконтроллерах. М.: Энергоатомиздат, 1990. 224 с.

Институт оптики атмосферы СО РАН,
Томск

Поступила в редакцию
13 декабря 1995 г.

A. P. Rostov. Versatile Many-channel AD Converter for Experimentation.

A version of 10-bit AD converter is described. It is furnished with RS-232 interface and equipment-compatible with any instrumentation used in international experiments.